TEMPLATE = app
RESOURCES += resources.qrc
QT += network declarative webkit

# libmultimediakit, from mobility:
CONFIG += mobility
MOBILITY = multimedia
ICON = icons/projecticon.svg

FORMS += \
    mainwindow.ui \
    sortdialog.ui \
    subscribedialog.ui
HEADERS = \
    playlistmodel.h \
    playlistlistmodel.h \
    mainwindow.h \
    metadatavalue.h \
    dirutils.h \
    metadataloader.h \
    sortdialog.h \
    subscriptionlistmodel.h \
    abstractmetadataloader.h \
    urlutils.h \
    completers.h \
    subscribedialog.h \
    episodes.h \
    downloadmgr.h \
    imageprovider.h
SOURCES = main.cpp \
    mainwindow.cpp \
    playlistmodel.cpp \
    playlistlistmodel.cpp \
    metadatavalue.cpp \
    dirutils.cpp \
    metadataloader.cpp \
    sortdialog.cpp \
    subscriptionlistmodel.cpp \
    abstractmetadataloader.cpp \
    urlutils.cpp \
    completers.cpp \
    subscribedialog.cpp \
    episodes.cpp \
    downloadmgr.cpp \
    imageprovider.cpp


symbian {
    LIBS += -lavkon -lcone -leikcore
    LIBS += -lremconcoreapi -lremconinterfacebase
    HEADERS += s60remote.h
    SOURCES += s60remote.cpp
    TARGET.EPOCSTACKSIZE = 0x14000
}

OTHER_FILES += todo.txt \
    README.txt \
    PlayListView.qml \
    SortOptions.qml \
    StyledText.qml \
    ParentListView.qml \
    subscribed.tsv \
    SubscriptionView.qml \
    TestFlickable.qml \
    FlickableViews.qml \
    MainMenu.qml \
    SongItem.qml \
    WrappedText.qml \
    index.html \
    m3uplayer-manual.qhp \
    m3uplayer-manual.qhcp \
    HelpDialog.qml \
    belle.qss


# On desktop only:

!symbian {
# taglib is recommended on windows...
#  works better than the mobility metadataloader on linux.
USE_TAGLIB=$$(USE_TAGLIB)
!isEmpty(USE_TAGLIB) {
    DEFINES += USE_TAGLIB
    LIBS += -ltag -L$$(TAGLIB_DIR)/lib
    SOURCES += tmetadataloader.cpp
    HEADERS += tmetadataloader.h
    INCLUDEPATH += $$(TAGLIB_DIR)/include
}

# optional object browser is from the libs of design patterns book
# http://www.distancecompsci.com/dist. For debugging purposes only.
USE_CPPLIBS=$$(CPPLIBS)
!isEmpty(USE_CPPLIBS) {
    DEFINES += USE_OBJECTBROWSER
    LIBS += -lobjectbrowser -L$$(CPPLIBS)
    INCLUDEPATH += $$(CPPLIBS)/objectbrowser
}

# place auto-generated files in "invisible" subdirectories
win32 {
    MOC_DIR = _moc
    UI_DIR = _ui
    OBJECTS_DIR = _obj
} else {
    UI_DIR = .ui
    MOC_DIR = .moc
    OBJECTS_DIR = .obj
}


}
